

Problem Set 8
Streams and Series

Issued: Monday, 23 October 2000

Due: Wednesday, 25 October 2000

Reading: Text (SICP 2nd Edition by Abelson & Sussman): Section 3.5

Using streams to represent power series

In Problem Set 6, you saw how to represent polynomials as lists of terms. In a similar way, we can work with *power series*, such as

$$\begin{aligned}e^x &= 1 + x + \frac{x^2}{2!} + \frac{x^3}{3!} + \frac{x^4}{4!} + \dots \\ \cos x &= 1 - \frac{x^2}{2!} + \frac{x^4}{4!} - \dots \\ \sin x &= x - \frac{x^3}{3!} + \frac{x^5}{5!} - \dots\end{aligned}$$

represented as streams of infinitely many terms. That is, the power series

$$a_0 + a_1x + a_2x^2 + a_3x^3 + \dots$$

will be represented as the infinite stream whose elements are $a_0, a_1, a_2, a_3, \dots$ ¹

Why would we want such a method? Well, let's separate the idea of a series representation from the idea of evaluating a function. For example, suppose we let $f(x) = \sin x$. We can separate the idea of evaluating f , e.g., $f(0) = 0, f(0.1) = 0.0998334$, from the means we use to compute the value of f . This is where the series representation is used, as a way of storing information sufficient to determine values of the function. In particular, by substituting a value for x into the series, and computing more and more terms in the sum, we get better and better estimates of the value of the function for that argument. This is shown in the table, where $\sin(1/10)$ is considered.

¹In this representation, all streams are infinite: a finite polynomial will be represented as a stream with an infinite number of trailing zeros.

<i>coefficient</i>	x^n	<i>term</i>	<i>sum</i>	<i>value</i>
0	1	0	0	0
1	$\frac{1}{10}$	$\frac{1}{10}$	$\frac{1}{10}$	0.1
0	$\frac{1}{100}$	0	$\frac{1}{10}$	0.1
$-\frac{1}{6}$	$\frac{1}{1000}$	$-\frac{1}{6000}$	$\frac{599}{6000}$	0.099833333333
0	$\frac{1}{10000}$	0	$\frac{599}{6000}$	0.099833333333
$\frac{1}{120}$	$\frac{1}{100000}$	$\frac{1}{12000000}$	$\frac{1198001}{12000000}$	0.09983341666

The first column shows the terms from the series representation for sine. This is the infinite series with which we will be dealing. The second column shows values for the associated powers of $1/10$. The third column is the product of the first two, and represents the next term in the series evaluation. The fourth column represents the sum of the terms to that point, and the last column is the decimal approximation to the sum.

With this representation of functions as streams of coefficients, series operations such as addition and scaling (multiplying by a constant) are identical to the basic stream operations. We provide series operations, though, in order to implement a complete power series data abstraction:

```
(define (add-streams s1 s2)
  (cond ((stream-null? s1) s2)
        ((stream-null? s2) s1)
        (else
         (cons-stream (+ (stream-car s1) (stream-car s2))
                       (add-streams (stream-cdr s1)
                                     (stream-cdr s2))))))

(define (scale-stream c stream)
  (stream-map (lambda (x) (* x c)) stream))

(define add-series add-streams)

(define scale-series scale-stream)

(define (negate-series s)
  (scale-series -1 s))

(define (subtract-series s1 s2)
  (add-series s1 (negate-series s2)))
```

You can use the following procedure to examine the series you will generate in this problem set:

```
(define (show-series s nterms)
  (if (= nterms 0)
      'done
      (begin (write-line (stream-car s))
              (show-series (stream-cdr s) (- nterms 1)))))
```

You can also examine an individual coefficient (of x^n) in a series using `series-coeff`:

```
(define (series-coeff s n)
  (stream-ref s n))
```

We also provide two ways to construct series. The procedure `coeffs->series` takes a list of initial coefficients and pads it with zeros to produce a power series. For example, `(coeff->series '(1 3 4))` produces the representation of the power series $1 + 3x + 4x^2 + 0x^3 + 0x^4 + \dots$.

```
(define (coeffs->series list-of-coeffs)
  (define zeros (cons-stream 0 zeros))
  (define (iter list)
    (if (null? list)
        zeros
        (cons-stream (car list)
                      (iter (cdr list)))))
  (iter list-of-coeffs))
```

`Proc->series` takes as argument a procedure p of one numeric argument and returns the series

$$p(0) + p(1)x + p(2)x^2 + p(3)x^3 + \dots$$

The definition requires the stream `non-neg-integers` to be the stream of non-negative integers: $0, 1, 2, 3, \dots$, which you will define in Exercise 3.

```
(define (proc->series proc)
  (stream-map proc non-neg-integers))
```

For example, `(proc->series square)` would return the stream $0, 1, 4, 9, 16, 25, \dots$

Pre-Code Exercises

Exercise 1: Without running the code, describe the elements of the streams defined by:

```
(define s1 (cons-stream 1 2))
(define s2 (cons-stream 1 (add-stream s2 s2)))
```

Exercise 2: Define a procedure `partial-sums` that takes as argument a stream s and returns the stream whose elements are $s_0, s_0 + s_1, s_0 + s_1 + s_2, \dots$. For example, `(partial-sums integers)` should be the stream $1, 3, 6, 10, 15, \dots$

Exercise 3: Also without executing the code, describe the streams produced by the following definitions. Assume that `integers` is the stream of positive integers (starting from 1):

```
(define x (cons-stream 1 (scale-stream 2 x)))

(define (mul-streams a b)
  (cons-stream
   (* (stream-car a) (stream-car b))
   (mul-streams (stream-cdr a)
                 (stream-cdr b))))

(define y (cons-stream 1 (mul-streams y integers)))
```

Lab Exercises

Load the code for Problem Set 8. **Note:** Loading the code for this problem set will change Scheme's basic arithmetic operations $+$, $-$, $*$, and $/$ so that they will work with rational numbers. For instance, `(/ 3 4)` will produce $3/4$ rather than 0.75 . You will find this useful in doing the exercises below.

Exercise 4: To get some additional practice with streams, write and turn in definitions for each of the following:

- **ones:** the infinite stream of 1s.
- **non-neg-integers:** the stream of integers 0, 1, 2, 3, 4, ...
- **alt-ones:** the stream 1, -1, 1, -1, ...
- **zeros:** the infinite stream of 0s. Your definition should use **alt-ones**, even though this may not be the most straightforward definition of **zeros**.

Now, show how to define the series:

$$\begin{aligned} S_1 &= 1 + x + x^2 + x^3 + \dots \\ S_2 &= 1 + 2x + 3x^2 + 4x^3 + \dots \end{aligned}$$

Turn in your definitions and a couple of coefficient printouts to demonstrate that they work.

Exercise 5: Multiplying two series is a lot like multiplying two multi-digit numbers, but starting with the left-most digit, instead of the right-most.

For example:

```

      11111
     x 12321
     -----
      11111
      22222
      33333
      22222
      11111
     -----
    136898631

```

Now imagine that there can be an infinite number of digits, *i.e.*, each of these is a (possibly infinite) series. (Remember that because each “digit” is in fact a term in the series, it can become arbitrarily large unlike in ordinary multiplication where carries are generated and propagated up to higher digit positions.)

Using this idea, complete the definition of the following procedure, which multiplies two series:

```

(define (mul-series s1 s2)
  (cons-stream <e1>
    (add-series <e2>
      <e3>)))

```

To test your procedure, demonstrate that the product of S_1 (from Exercise 4) and S_1 is S_2 . What is the coefficient of x^{10} in the product of S_2 and S_2 ? Turn in your definition of **mul-series**. (Optional: Give a general formula for the coefficient of x^n in the product of S_2 and S_2 .)

Exercise 6: Inverting a power series Let S be a power series whose constant term is 1. We will call such a power series a *unit power series*. Suppose we want to find the *inverse* of S , namely, the power series X such that $S \cdot X = 1$. To see how to do this, write $S = 1 + S_R$ where S_R is the rest of S after the constant term. Then we want to solve the equation $S \cdot X = 1$ for S and we can do this as follows:

$$\begin{aligned} S \cdot X &= 1 \\ (1 + S_R) \cdot X &= 1 \\ X + S_R \cdot X &= 1 \\ X &= 1 - S_R \cdot X \end{aligned}$$

In other words, X is the power series whose constant term is 1 and whose rest is given by the negative of S_R times X .

Use this idea to write a procedure `invert-unit-series` that computes $1/S$ for a unit power series S . To test your procedure, invert the series S_1 (from Exercise 4) and show that you get the series $1 - x$. (Convince yourself that this is the correct answer.) Turn in a listing of your procedure. This is a very short procedure, but it is very clever. In fact, to someone looking at it for the first time, it may seem that it can't work—that it must go into an infinite loop. Write a few sentences of explanation explaining why the procedure does in fact work, and does not go into a loop.

Exercise 7: Use your answer from Exercise 6 to produce a procedure `div-series` that divides two power series. `Div-series` should work for any two series, provided that the denominator series begins with a non-zero constant term. (If the denominator has a zero constant term, then `div-series` should signal an explanatory error.) Turn in a listing of your procedure along with three or four well-chosen test cases (and demonstrate why the answers given by your division are indeed the correct answers).

Exercise 8: Now suppose that we want to integrate a series representation. By this, we mean that we want to perform symbolic integration, thus, for example, given a series

$$a_0 + a_1x + a_2x^2 + a_3x^3 + \dots$$

we want to return the integral of the series (except for the constant term)

$$a_0x + \frac{1}{2}a_1x^2 + \frac{1}{3}a_2x^3 + \frac{1}{4}a_3x^4 + \dots$$

Define a procedure `integrate-series-tail` that will do this. Note that all you need to do is transform the series

$$a_0, \quad a_1, \quad a_2, \quad a_3, \quad a_4, \quad a_5, \quad \dots$$

into the series

$$a_0, \quad \frac{a_1}{2}, \quad \frac{a_2}{3}, \quad \frac{a_3}{4}, \quad \frac{a_4}{5}, \quad \frac{a_5}{6}, \quad \dots$$

Note that this means that the procedure generates the coefficients of a series starting with the first order coefficient, not that the zeroth order coefficient is 0.

Turn in a listing of your procedure and demonstrate that it works by computing `integrate-series-tail` of the series S_2 from Exercise 4.

Exercise 9: Demonstrate that you can generate the series for e^x as

```
(define exp-series
  (cons-stream 1 (integrate-series-tail exp-series)))
```

Explain the reasoning behind this definition. Show how to generate the series for sine and cosine, in a similar way, as a pair of mutually recursive definitions. It may help to recall that

$$\int \sin x = -\cos x$$

and

$$\int \cos x = \sin x.$$

(The power series for sin and cos were given in the introduction to the problem set.)

Exercise 10: One very useful feature of a power series representation for a function is that one can use the initial terms in the series to get approximations to the function. For example, suppose we have

$$f(x) = a_0 + a_1x + a_2x^2 + a_3x^3 + \dots$$

We have represented this by the series of coefficients:

$$a_0, \quad a_1, \quad a_2, \quad a_3, \quad \dots$$

Now suppose that we want to approximate the value of the function f at some point x_0 . We could successively improve this approximation² by considering the following

$$f(x_0) \approx a_0 \tag{1}$$

$$f(x_0) \approx a_0 + a_1x_0 \tag{2}$$

$$f(x_0) \approx a_0 + a_1x_0 + a_2x_0^2 \tag{3}$$

Notice that each of these expressions (1–3) could also be captured in a stream representation, with first term a_0 , second term $a_0 + a_1x_0$ and so on.

Implement this idea by defining a procedure `approximate` which takes as arguments a value x_0 and a series representation of a function f , and which returns a stream of successive approximations to the value of the function at that point $f(x_0)$. Turn in a listing of your code, as well as examples of using it to approximate some functions. Note that to be very careful, this is not a series representation but a stream one, so you may want to think carefully about which representations to use.

²Actually, there are some technical issues about whether the argument is in the radius of convergence of the series, but we will ignore that issue here.